We load a package for the computation of Formal Power Series which is based on the given algorithm [Koepf (1992)] and was written jointly with Dominik Gruntz.

```
> read "FPS.mpl";
```

Package Formal Power Series, Maple V-8

Copyright 1995, Dominik Gruntz, University of Basel

Copyright 2002, Detlef Müller & Wolfram Koepf, University of Kassel

As a simple exercise we compute the power series of the Koebe function.

```
> FPS(z/(1-z)^2,z);
```

$$\sum_{k=0}^{\infty} (k+1)\, z^{(k+1)}$$

As a more complicated exercise we compute the power series of the *k*th power Loewner chain of the Koebe function. (y=E^(-*t*)).

```
> w:=4*y*z/(1-z+sqrt(1-2*(1-2*y)*z+z^2))^2;
```

$$w := \frac{4\, y\, z}{\left(1 - z + \sqrt{1 - 2\, z + 4\, z\, y + z^2}\,\right)^2}$$

```
> assume(z>0,z<1); interface(showassumed=0):
> y^k*FPS((w/y)^k,y,j);
FPS/hypergeomRE:   provided that    -1    <=    min(-1,-1-2*k)
```

$$y^k \left(\sum_{j=0}^{\infty} \frac{(-1)^j\, z^{(j+k)} \left(\frac{1}{(z-1)^2}\right)^{(j+k)} \mathrm{pochhammer}(2\,k, 2\,j)\, y^j}{\mathrm{pochhammer}(1+2\,k, j)\, j!} \right)$$

How does the algorithm work?

```
> infolevel[FPS]:=5:
> FPS((w/y)^k,y,j);
FPS/FPS:   looking for DE of degree   1
FPS/FPS:   looking for DE of degree   2
FPS/FPS:   DE of degree   2   found.
FPS/FPS:   DE =
```

$$(y - 2\, z\, y + 4\, y^2\, z + z^2\, y)\, F''(x) + (6\, z\, y + 8\, k\, z\, y + z^2 + 2\, k\, z^2 - 2\, z - 4\, k\, z + 1 + 2\, k)\, F'(x) + (2\, k\, z + 4\, k^2\, z)\, F(x) = 0$$

```
FPS/FPS:   RE =
```

$$a(j+1) = -\frac{2\, z\, (2\, j + 1 + 2\, k)\, (j + k)\, a(j)}{(z-1)^2\, (j+1)\, (j + 1 + 2\, k)}$$

```
FPS/hypergeomRE:   RE is of hypergeometric type.
FPS/hypergeomRE:   Symmetry number m :=    1
FPS/hypergeomRE:   RE:
```

$$(z-1)^2\, (j+1)\, (j + 1 + 2\, k)\, a(j+1) = -2\, (j + k)\, (2\, j + 1 + 2\, k)\, z\, a(j)$$

```
FPS/hypergeomRE:   provided that    -1    <=    min(-1,-1-2*k)
FPS/hypergeomRE:   RE valid for all k >=    0
FPS/hypergeomRE:   a(0) =   (z/(z-1)^2)^k
```

$$\sum_{j=0}^{\infty} \frac{(-1)^j z^{(j+k)} \left(\frac{1}{(z-1)^2}\right)^{(j+k)} \text{pochhammer}(2\,k, 2\,j)\, y^j}{\text{pochhammer}(1+2\,k, j)\, j!}$$

```
> infolevel[FPS]:=0:
>
```

Representation of the Weinstein functions

```
> TIME:=time():
reihe1:=y^(k+1)*FPS(1/y*(w/y)^(k+1)/(1-w^2),y,j);
time()-TIME;
```

$$reihe1 := \frac{y^{(1+k)} \left(\sum_{j=0}^{\infty} \frac{(-1)^j z^{(j+k+1)} \left(\frac{1}{(z-1)^2}\right)^{(j+k)} \text{pochhammer}(1+2\,k, 2\,j)\, y^{(j-1)}}{\text{pochhammer}(1+2\,k, j)\, j!} \right)}{(z-1)^2}$$

$$25.707$$

```
> reihe2:=FPS(reihe1,z,i);
```

$$reihe2 := \sum_{j=0}^{\infty} \left(\sum_{i=0}^{\infty} z^{(j+k+1)} \left(\frac{2\,(-1)^j z^i \,\text{pochhammer}(2+2\,k+2\,j, i)\, \Gamma(2\,j+2\,k)\, y^{(j+k)}\, j}{i!\,(j+2\,k)\, j!\, \Gamma(j+2\,k)} + \frac{2\,(-1)^j z^i y^{(j+k)}\, k\, \text{pochhammer}(2+2\,k+2\,j, i)\, \Gamma(2\,j+2\,k)}{\Gamma(j+2\,k)\,(j+2\,k)\, j!\, i!} \right) \right)$$

```
> summand:=op([1,1],reihe2);
```

$$summand := z^{(j+k+1)} \left(\frac{2\,(-1)^j z^i \,\text{pochhammer}(2+2\,k+2\,j, i)\, \Gamma(2\,j+2\,k)\, y^{(j+k)}\, j}{i!\,(j+2\,k)\, j!\, \Gamma(j+2\,k)} + \frac{2\,(-1)^j z^i y^{(j+k)}\, k\, \text{pochhammer}(2+2\,k+2\,j, i)\, \Gamma(2\,j+2\,k)}{\Gamma(j+2\,k)\,(j+2\,k)\, j!\, i!} \right)$$

```
> summand:=simplify(subs(i=n-j-k,summand));
```

$$summand := \frac{(-1)^j y^{(j+k)} z^{(n+1)}\, \Gamma(2+k+j+n)}{(2\,j+1+2\,k)\, \Gamma(j+1+2\,k)\, \Gamma(n-j-k+1)\, \Gamma(j+1)}$$

```
> result:=convert(sumtools[sumtohyper](summand,j),binomial);
```

$$result := y^k z^{(n+1)}\, \text{hypergeom}\left(\left[k-n, 2+k+n, \frac{1}{2}+k \right], \left[1+2\,k, \frac{3}{2}+k \right], y \right) \text{binomial}(1+k+n, 1+2\,k)$$

```
>
```